
Zusammenfassung 

SAS bietet seit der Version 9.4 die Möglichkeit, mit dem Output Delivery System, kurz 
ODS, native Excel Dateien zu erzeugen. Der neue ODS Ausgabekanal heißt ODS Excel, 
wurde mit dem SAS Release 9.4 (TS1M1) von SAS als „Experimental Feature“ eingeführt 
und erlangte mit dem SAS Release 9.4 (TS1M3) den Status „Full Supported“. 
In diesem Beitrag werden die folgenden Themen behandelt: 

- Übersicht der Eigenschaften von Tageset ExcelXP – ODS Excel – DDE 
- Einführung in ODS Excel – Aktionen –  Optionen – Unteroptionen 
- Die Option: Contents, #Byval, ID 
- Unteroptionen mit Alias 
- Metadaten Option 
- Formatieren der Ausgabe 
- Excel Formeln 
- CSS Cascading Style Sheet 

 
Schlüsselwörter: PROC TEMPLATE, ODS EXCEL, CSS, TAGSETS.EXCELXP, 
PROC REPORT, PROC SGPLOT, PROC GLM, PROC MEANS, DDE 

1 Einführung in ODS Excel 

1.1 Tageset ExcelXP – ODS Excel – DDE 

In der folgenden Tabelle werden die wichtigsten Eigenschaften des Tagset ExcelXP, des 
ODS Ausgabekanals Excel und der Methode Dynamic Data Exchange, kurz DDE, ge-
genübergestellt. 
 
Tabelle 1: Tageset ExcelXP – ODS Excel – DDE 
Eigenschaften ExcelXP (1) ODS Excel (2) DDE 
Pivot Tabelle Ja Nein Benötigt Excel Makro 
Filter Ja Ja Ja 
Title Ja Ja Ja 
Auto. Anpassung 
von Spalten 

Nein Ja Ja 

Einfrieren Ja Ja Ja 

Grafiken Nein Ja Ja 
Mehrere Blätter Ja Ja Ja 
Kopf- Fußzeile Ja Ja Ja 
System unabhängig Ja Ja Nein - nur Windows 
Dateiendung xml xlsx xls/xlsx 

(1) ExcelXP fällt in die Kategorie ‚tagsets‘. Es ist kein fest eingebauter ODS Ausgabekanal und 
kann vom SAS Benutzer modifiziert und erweitert werden. 

(2) Verfügbar ab SAS Version 9.4 – Experimental Version 9.4 (TS1M1) – Full Support 9.4 
(TS1M3) 

1.2 ODS Ausgabekanäle 

ODS Excel ist ein standardisierter 
ODS Ausgabekanal und somit auf dem 
gleichen Level wie ODS RTF, ODS 
PDF, ODS HTML usw. 
 

 

 Abbildung 1: ODS Ausgabekanäle 

1.3 ID und Aktionen 
ods excel <( < id = > identifier) > < aktion >; 
ods excel <( < id = > identifier) > < option(en) >; 
Die ID Feature ermöglicht das Erstellen mehrerer Excel Arbeitsmappen oder anderer 
ODS Dateien zum Beispiel PDF, RTF oder HTML in einem laufenden Prozess. 
 
Tabelle 2: Mögliche ODS Excel Aktionsparameter 
CLOSE Schließt ein ODS Excel Statement mit oder ohne ID Option 
EXCLUDE Der ODS EXCLUDE Befehl unterbindet die Ausgabe eines 

oder mehrere ODS Objektes 
SELECT Der ODS SELECT Befehl fügt ein oder mehrere ODS Objekte 

in die Ausgabedatei ein 
SHOW Der ODS Show Befehl gibt eine Liste der selektierten oder aus-

geschlossenen Werte in der SAS LOG zurück 
NONE/ALL 
(SELECT/EXCLUDE) 

ALL = sendet alle erstellten Ausgaben - 
NONE = sendet keine der erstellten Ausgaben - 
      zum SAS Default Ausgabeverzeichnis 

ODS TRACE Erzeugt in der SAS LOG eine Übersicht der möglichen ODS 
Ausgabeobjekte – kann mit ODS SELECT/EXCLUDE gesteu-
ert werden 

 
ODS TRACE LOG AUSZUG: 
ods trace on / label; 
 
proc univariate data=out.statepop mu0=3.5; 
        var citypop_90 citypop_80; 
run; 
 
Ausgabe hinzugefügt: 
------------- 
Name:         Moments 
Etikett:      Momente 
Template:     base.univariate.Moments 
Pfad:         Univariate.CityPop_90.Moments 
Etikettpfad:  'Proc Univariate'.'CityPop_90'.'Momente' 
 ------------- 
 
Ausgabe hinzugefügt: 
------------- 
Name:BasicMeasures 
Etikett:      Lage- und Streuungsmaße 
Template:     base.univariate.Measures 
Pfad:         Univariate.CityPop_90.BasicMeasures 
Etikettpfad:  'Proc Univariate'.'CityPop_90'.'Lage- und 
Streuungsmaße' 
------------- 

1.4 Optionen – Übersicht 

Tabelle 3: ODS Excel Optionen 
ANCHOR= AUTHOR= BOX_SIZING= 
CATEGORY= COMMENTS= CSSSTYLE= 
DOM= DPI= FILE= 
GFOOTNOTE NOGFOOTNOTE GTITLE 
NOGTITLE IMAGE_DPI= KEYWORDS= 
ID= OPTIONS SASDATE 
STATUS= STYLE= TEXT= 
TITLE= WORK=  
 
Bis auf die Optionen GFOOTNOTE/NOGFOOTNOTE und GTITLE/NOGTITLE wer-
den Parameter akzeptiert. Bei den meisten ODS Excel Optionen handelt es sich um 
Standard ODS Optionen, die auch bei anderen ODS Ausgabekanälen angewendet wer-
den können. Nähere Erklärung zu den Optionen DOM und ANCHOR: 
DOM steht für Document Object Model, wenn diese Option gesetzt wird erstellt SAS in 
der SAS LOG oder in einer externen Datei HTML Code. DOM ist eine in-memory Dar-

stellung des erstellten ODS Reports. Die ANCHOR Option funktioniert ausschließlich 
mit der Option DOM, mit ihr können Sektionen in der DOM Ausgabe benannt werden. 

Beispiel Code: 
ods excel file="example001.xlsx" anchor='KSFE2018'  
          dom="example001.htm"; 
    proc print data=sashelp.cars(where=(upcase(make)='AUDI')obs=5)  
          noobs; 
    run; 
ods excel close; 

DOM Dokumente 
Code mit der 
ANCHOR Option 
<Sektion ID> 

 

 Abbildung 2: DOM Dokument Code 

1.5 Unteroptionen von den Optionen – Übersicht 

Tabelle 4: ODS Excel Unteroptionen von den Optionen 
ABSOLUTE_COLUMN_WIDTH
= 

ABSOLUTE_ROW_HEIGHT
= 

AUTOFILTER= 

BLACKANDWHITE= BLANK_SHEET= CENTER_HORIZONTAL= 
CENTER_VERTICAL= COLUMN_REPEAT= CONTENTS= 
DPI= DRAFTQUALITY= EMBEDDED_FOOTNOTES= 
EMBED_FOOTNOTES_ONCE EMBEDDED_TITLES= EMBED_TITLES_ONCE= 
FITTOPAGE= FORMULAS= FROZEN_HEADERS= 
FROZEN_ROWHEADERS= GRIDLINES= HIDDEN_COLUMNS= 
HIDDEN_ROWS= INDEX= MSG_LEVEL= 
ORIENTATION= PAGE_ORDER_ACROSS= AGES_FITHEIGHT= 
PAGES_FITWIDTH= PRINT_AREA= PRINT_FOOTER= 
PRINT_FOOTER_MARGIN= PRINT_HEADER= PRINT_HEADER_MARGIN= 
ROWBREAKS_COUNT= ROWBREAKS_INTERVAL= ROWCOLHEADINGS= 
ROW_HEIGHTS= ROW_REPEAT= SCALE= 
SHEET_LABEL= SHEET_NAME= START_AT= 
SUPPRESS_BYLINES= TAB_COLOR= TITLE_FOOTNOTE_NOBREAK

= 
TITLE_FOOTNOTE_WIDTH= ZOOM= FLOW= 
PROTECT_WORKSHEET= SHEET_INTERVAL=  

1.6 Der Programmrahmen 
ods excel file="example01.xlsx" style=harvest; 
    proc print data=sashelp.cars(where=(upcase(make)='AUDI')) noobs; 
      by type; 
    run; 
ods excel close; 

Durch die BY Anweisung in der PRINT Prozedur wird für jeden Eintrag der variablen 
type aus der Datei SASHELP.CARS ein neues Excel Arbeitsblatt erstellt. 

Einfache PROC 
PRINT Ausgabe. 
Verwendet wird 
der ODS Style 
HARVEST. 

 

 Abbildung 3: ODS Excel Output 1 

2 ODS Excel – Optionen 

2.1 Die Option Contents 
ods excel file="example02.xlsx" options(contents='on')   
    style=harvest; 
    proc print data=sashelp.cars(where=(upcase(make)='AUDI')) noobs; 
         by type; 
    run; 
ods excel close; 

Mit der Option contents=‘on‘ oder contents=‘yes‘ erstellt ODS EXCEL ein separates 
Arbeitsblatt mit einem Inhaltsverzeichnis der SAS Prozedurausgaben. 

ODS EXCEL 
Ausgabe mit In-
haltsverzeichnis 

 Abbildung 4: ODS Excel Output 2 

2.2 Häufig verwendete Optionen 

Programm 1 
ods excel file="example03.xlsx" 
    options(contents='on' 
    blank_sheet='KSFE 2018 - leer' 
    embedded_footnotes='on' 
    embedded_titles='on' 
    sheet_name='KSFE 2018 - example03' 
    frozen_headers='on' 
    frozen_rowheaders='on' 
) 
style=harvest; 
 
title1 'KSFE 2018 - title 1'; 
footnote1 'KSFE 2018 - Fussnote 1'; 
 
    proc print data=sashelp.cars(where=(upcase(make)='AUDI')) noobs; 
         by type; 
   run; 
 
ods excel close; 

Erklärung der verwendeten Optionen: 
· BLANK_SHEET =‘KSFE 2018 – leer‘ erstellt ein leeres Arbeitsblatt mit dem 

Namen KSFE 2018 – leer 
· EMBEDDED_FOOTNOTES = ‘ON‘ Fußnoten sind Teil des Arbeitsblattes 
· EMBEDDED_TITLES = ‘ON’ Titel sind Teile des Arbeitsblattes 
· SHEET_NAME = ‘KSFE 2018 – example03‘ ist der Name des Arbeitsblattes 
· FROZEN_HEADERS = ‘ON‘ der Kopf (Title, BY Information) des Arbeitsblat-

tes wird dadurch eingefroren 
· FROZEN_ROWHEADERS = ‘ON‘ die Zeile mit den variablen Namen wird 

dadurch eingefroren 

ODS EXCEL 
Ausgabe 3 

 
 Abbildung 5: ODS Excel Output 3 

Programm 2 
ods excel file="example04.xlsx"  
    options(embedded_footnotes='on' 
    embed_footnotes_once='on' 
    embedded_titles='on' 
    embed_titles_once='on' 
    title_footnote_nobreak='on'  
    sheet_name='Cars DS Zusammenfassung' 
    sheet_interval='none' 
) 
style=pearl; 
 
title1 'KSFE 2018 - Cars Datensatz Zusammenfassung und Histogramm'; 
footnote1 'KSFE 2018 - Ich bin eine Fußnote'; 
 
ods escapechar='~'; 
ods text="~S={font_size=10pt font_weight=bold}~Cars DS    
    Zusammenfassung und Histogramm"; 
 
proc means data=sashelp.cars; 
     var msrp invoice; 
run; 
 
ods graphics / height=400 width=800 noborder; 
proc sgplot data=sashelp.cars; 
     histogram msrp; 
run; 
 
ods excel close; 

 
Mit dem Statement ODS TEXT wird eine Zeile mit dem Text Zusammenfassung und 

Histogramm in das Arbeitsblatt geschrieben 
 
Erklärung der verwendeten Optionen: 

· EMBED_FOOTNOTES_ONCE = ‘ON‘ die Fußnote wird nur einmal ausgegeben und nicht 
wie normalerweise nach jeder Prozedur. 

· EMBED_TITLES_ONCE = ‘ON‘ der Titel wird nur einmal ausgegeben und nicht, wie üblich 
nach jeder Prozedur. 

· TITLE_FOOTNOTE_NOBREAK = ‘ON‘ kein Umbruch in der Fußnote oder im Titel. 
· SHEET_INTERVAL = ‘NONE‘ die Prozedur Ausgabe wird in ein Arbeitsblatt geschrieben. 

Weitere Werte: BYGROUP = erstellt nach jeder BY Gruppe ein neues Arbeitsblatt, NOW = er-
stellt nach jedem neuen Output ein neues Arbeitsblatt, PAGE = erstellt ein neues Arbeitsblatt 
für eine Seite (abhängig von der Prozedur), PROC = erstellt ein neues Arbeitsblatt für jeden 
Prozeduraufruf, TABLE = erstellt für jede neue Tabelle ein neues Arbeitsblatt. 

ODS EXCEL 
Ausgabe 4 

 
 Abbildung 6: ODS Excel Output 4 

2.3 Die #BYVAL Funktion 
ods excel file=" example10.xlsx" 
    options(sheet_name="#byval(age)"); 
 
    proc print data=work.class noobs; 
         by age; 
    run; 
 

ods excel close; 
 
Die #BYVAL 
Funktion erstellt ein 
neues Arbeitsblatt so-
bald sich der Wert der 
Variablen ändert. 

 

 Abbildung 7: ODS Excel Output 5 

2.4 Excel Metadaten Optionen 
ods excel file="example05.xlsx"  
    author='Frank Biedermann' 
    category='KSFE 2018 ods' 
    comments='Beispiel Programm für die KSFE 2018' 
    keywords='KSFE 2018 ods excel metadata' 
    status='Draft' 
   title='KSFE 2018 example 05' 
    sasdate 
    style=harvest; 

    proc print data=sashelp.cars(where=(upcase(make)='AUDI')) noobs; 
         by type; 
    run; 

ods excel close; 

 
Erklärung der verwendeten Optionen: 

· AUTHOR = ‘Frank Biedermann‘ der Excel Metadatenwert Autor wird hierdurch geändert. 
· CATEGORY = ‘KSFE 2018 ods‘ der Excel Metadatenwert Kategorien wird hierdurch geän-

dert. 
· COMMENTS = ‘Beispiel Programm für die KSFE 2018‘ der Excel Metadatenwert Kommenta-

re wird hierdurch geändert. 
· KEYWORDS = ‘KSFE 2018 ods excel metadata‘ der Excel Metadatenwert Tags wird hier-

durch geändert. 
· STATUS = ‘Draft‘ der Excel Metadatenwert Status wird dadurch geändert. 
· TITLE = ‘KSFE 2018 example 05‘ der Excel Metadatenwert Title wir hierdurch geändert. 
· SASDATE schreibt das SAS Session Start Datum/Uhrzeit in die Excel Dateimetadaten. 

ODS EXCEL 
Ausgabe 6 

 

 Abbildung 8: ODS Excel Output 6 

3 Die Sachen mit dem Alias 
Die Unteroptionen in der Tabelle erlauben die Verwendung von Alias-Namen. 
 
Tabelle 5: ODS Excel Unteroptionen – Alias 
BLACKANDWHITE=on – alias – yes, 
off (default) – alias – no   

CENTER_HORIZONTAL=on – alias – 
yes, off (default) – alias – no 

CENTER_VERTICAL=on – alias – yes, 
off (default) – alias – no 

DRAFTQUALITY=on – alias – yes, off 
(default) – alias – no 

EMBEDDED_FOOTNOTES=on – alias 
– yes, off (default) – alias – no 

EMBED_FOOTNOTES_ONCE 
alias EMBED_FOOTERS_ONCE =on – 
alias – yes, off (default) – alias – no 

EMBEDDED_TITLES=on – alias – yes, 
off (default) – alias – no 

EMBED_TITLES_ONCE=on – alias – 
yes, off (default) – alias – no 

FITTOPAGE=on – alias – yes, off (de-
fault) – alias – no 

FLOW=data, headers – alias – header, 
rowheaders – alias – rowheader, tables – 
alias table, text 

FORMULAS=on – alias – yes, off (de-
fault) – alias – no 

FROZEN_HEADERS=on – alias – 
yes|true, off (default) – alias – no|false 

FROZEN_ROWHEADERS=on – alias 
– yes, off (default) – alias – no 

GRIDLINES=on – alias – yes, off (de-
fault) – alias – no 

INDEX=on – alias – yes, off (default) – 
alias – no  

PAGE_ORDER_ACROSS=on – alias – 
yes, off (default) – alias – no  

PROTECT_WORKSHEET=on – alias – 
yes, off (default) – alias – no 

ROWCOLHEADINGS=on – alias – yes, 
off (default) – alias – no 

SHEET_INTERVAL=bygroup – alias – 
bygroups, none, now, page, proc, table – 
alias – output 

SUPPRESS_BYLINES=on – alias – yes, 
off (default) – alias – no  

TITLE_FOOTNOTE_NOBREAK=on – 
alias – yes, off (default) – alias – no 

 

4 Die ID Option 
ods excel file=" example06_kein_style.xlsx"; 
options orientation=landscape; 
ods excel (id=pearl) 
    style=pearl 
    file="example06_id_peral_style.xlsx"; 
options orientation=portrait; 
ods rtf (id=minimal) 
    style=minimal 
    file="example06_id_minimal_style.rtf"; 
    options orientation=landscape; 
ods pdf (id=sasweb) 
    style=sasweb 
    file="example06_id_sasweb_style.pdf" notoc; 

    proc print data=sashelp.cars(where=(upcase(make)='AUDI')) noobs; 
         by type; 
   run; 
 
ods excel close; 
ods excel (id=pearl) close; 
ods rtf (id=minimal) close; 
ods pdf (id=sasweb) close; 

 
Die ID (Identifier) Option erlaubt das Öffnen mehrerer ODS Kanäle. Des Weiteren kön-
nen mit Hilfe der ID Option, unterschiedliche Optionen und Unteroptionen, den ver-
schiedenen ODS Ausgaben zugewiesen werden. 
 
ODS EXCEL 
Ausgabe 7 

 

 Abbildung 9: ODS Excel Output 7 

5 Ausgabe formatieren 

5.1 Template und Optionen 

Programm 1 – Grid Ausgabe 

SAS Benutzer, welche kein SAS 
Access Interface to PC Files lizen-
siert haben, können auch klassische 
Excel Arbeitsmappen erstellen (Grid 
Ausgabe) 

 

 Abbildung 10: ODS Excel Output 8 

proc template; 
     define style styles.grid; 
        parent=styles.excel; 
        class header / background=_undef_ 
        color=black; 
        class body / background=_undef_; 
     end; 
run; 

Mit der Prozedur TEMPLATE erstellen wir einen neuen Style mit dem Namen grid, als 
Vorlage dient der Style Excel. Der Hintergrund (background) wird beim header und bo-
dy auf _undef_ gesetzt. 

ods excel file="example07.xlsx" 
    options(sheet_name='Datensatz class') 
    style=styles.grid; 

    proc print data=sashelp.class noobs; 
    run; 

ods excel close; 

Die ODS EXCEL Ausgabe muss anschließend auf den neuen Style referenziert werden. 

Programm 2 – Langer Text 
 
Für die Ausgabe von langen Texten in Excel Arbeitsmappen, bringt ODS Excel einige 
Optionen mit, welche noch nicht optimal funktionieren. 
 
Im Beispiel werden folgende Optionen verwendet: 

- absolute_row_height 
- absolute_column_width 
- flow 

 
data work.cars; 
     set sashelp.cars; 
     length longtxt $250.; 
     longtxt='AaaaaaaaaaaaaaaaaaaaaaaaaaaBbbbbbbbbbbbbbbbbbbbbb 
     CccccccccccccccccccccccccDddddddddddddddddddddddEeeeeeeeeeeeee 
     eeeeeeee'; 
run; 
 
ods excel file="example08.xlsx" 
    options(absolute_row_height='25' 
       absolute_column_width='10, 10, 100' 
       flow='C2:C20') 
    style=sasweb; 
 
proc print data=work.cars(obs=20) noobs; 
     var make model longtxt; 
run; 
ods excel close; 

 
Die Option FLOW wurde 
auf der Variablen longtxt 
angewendet, das Ergebnis ist 
nicht optimal, beim Verklei-
nern der Spalte C wird der 
Text nicht richtig umgebro-
chen. 

 

 Abbildung 11: ODS Excel Output 9 

Eine Empfehlung hier, ist die Verwendung von Style Elementen aus dem 
tagsets.excelxp, zusammen mit der ODS Excel flow Option. 
Hier ein Beispiel mit der REPORT Prozedur: 
 
ods excel file="example08.xlsx" 
    options(flow='E2:E20') 
    style=sasweb; 
 
proc report data=work.cars(obs=20) nowd missing 
     style(column)=[vjust=bottom just=left] 
     style(header)=[vjust=middle]; 
     columns make model type origin longtxt; 
       define make    / style(column)=[width=50pt]  'Make'; 
       define model   / style(column)=[width=70pt]  'Model'; 
       define type    / style(column)=[width=50pt]  'Type';  
       define origin  / style(column)=[width=50pt]  'Origin'; 
      define longtxt / style(column)=[width=120pt] 'Long Text'; 
run; 
 
ods excel close; 

 
Nicht Optimal aber 
besser. Des Weiteren 
konnte die Ausgabe 
durch REPORT Opti-
onen verbessert wer-
den. 

 Abbildung 12: ODS Excel Output 10 

5.2 CSS 

ODS Excel erlaubt die Verwendung von Cascading Style Sheets (CSS), diese werden 
mit der Option cssstyle eingebunden und ersetzen den ODS Style. 
 
filename css "css_style.css"; 
 
data _null_; 
     file css noprint linesize=132; 
     put ".body {background-color: lightblue;    } "/ 
         ".header{background-color: yellow;         " / 
             "font-family: Arial, Helvetica, sans-serif;"/ 

             "font-weight: bold;} " / 
         ".rowheader {background-color: darkblue;"/ 
             "color: white;} " / 
         ".data {background-color: lightred;" / 
             "font-family: Arial, Helvetica, sans-serif; }"; 
run; 

ods excel file="example11.xlsx"  
options(sheet_name="CSS - Format") 
cssstyle="css_style.css"; 

proc print data=sashelp.class noobs; 
run; 

ods excel close; 

 
Excel Ausgabe mit CSS Style  

 Abbildung 13: ODS Excel Output 11 

6 Formeln 
ODS Excel ermöglicht es ebenfalls Excel Formeln in einem Arbeitsblatt auszugeben. 
Der folgende Beispiel Code zeigt wie Sie A1 Notation und R1C1 (Relative) Notation 
Formeln mit der REPORT Prozedur ausgegeben werden können. 

A1 Notation = Spalten 
Name und Zeilennummer 

 

 Abbildung 14: A1 Notation 

R1C1 Notation auch Rela-
tive Notation 

 
 Abbildung 15: R1C1 Notation 

options missing=0; 
ods excel file="example09.xlsx" 
    style=styles.grid; 

   proc report data=sashelp.prdsale spanrows; 
      col Region Product Prodtype Actual Predict Difference; 
      define region / group style(column)={vjust=middle 
         tagattr="rotate:90" just=center}; 
      define product / group; 
      define prodtype/ group; 
      define difference / computed 
         style(column)={tagattr="format:[color15]$#,### formula:RC[- 
         1]-RC[-2]"}; 
      compute after; 
         region="Total"; 
      endcomp; 
      compute predict; 
        if upcase(_break_)='_RBREAK_' then do; 
          call define(_col_,"style","style={ 
             tagattr=""formula:=sum($E$2:$E11)""}"); 
          call define('actual.sum',"style","style={ 
             tagattr=""formula:=sum($D$2:$D11)""}"); 
          call define('region',"style","style={ 
             color=red tagattr=""rotate:0""}"); 

        end; 
      endcomp; 
      rbreak after / summarize; 
   run; 

ods excel close; 

Excel Ausgabe mit 
Formeln. 
A1 Notation Formel 
in unserem Beispiel 
Summe 
($D$2:$D11) und 
($E$2:$E11) 
R1C1 Notation auch 
Relative Notation in 
unserem Beispiel 
(=E2-D2) , (=E3-
D3) …. 

 

 Abbildung 16: ODS Excel Output 12 

